

8 . 変数変換を使った積分

Maple での積分計算は、記号（公式を使った）積分と数値積分の両方が用意されています。

◆ 記号積分

int コマンドは記号積分を行います。定積分と不定積分の両方を扱えます。定積分の場合は、その範囲を指定します。

```
[> restart;
不定積分
> int(sin(x),x);
                                -cos(x)
(1)
```

```
定積分
> int(sin(x),x=-Pi..Pi);
                                0
(2)
```

◆ 数値積分

数値積分を行う場合、Int コマンドの結果を evalf コマンドで評価し数値化します。

```
[> evalf(Int(sin(x),x=-Pi..Pi));
                                0.
(3)
```

Int コマンドは int と同じく積分を計算するコマンドですが、その場で積分計算せずに計算式をそのまま表示します。評価を後回しにする意味を持ちます。

この他、Diff と diff、Limit と limit などおなじような関係のコマンドがあります。また、Maple は int コマンドで結果が得られなかった場合、Int の形で結果を返します。これをそのまま evalf に適用すると、数値積分結果が得られます。

```
[> int(1/GAMMA(x),x=0..2);
                                 $\int_0^2 \frac{1}{\Gamma(x)} dx$ 
(4)
```

```
> evalf((4));
                                1.626378399
(5)
```

◆ 二重積分

二重積分を行う場合は、int コマンドを重ねて使います。

```
[> int(int(exp(-x^2-y^2),x),y);
                                 $\frac{1}{4} \pi \operatorname{erf}(x) \operatorname{erf}(y)$ 
(6)
```

数値積分も同様に計算します。

```
[> Int(Int(exp(-x^2-y^2),x=-3..3),y=-3..3);
                                 $\int_{-3}^3 \int_{-3}^3 e^{-x^2-y^2} dx dy$ 
(7)
```

```
> evalf((7));
                                3.141453856
(8)
```

◆事例

$e^{-x^2 - \frac{y^2}{5}}$ を半径 2 の円柱で切り取った部分の体積を求めます。
 この計算は例えば、ある光源の瞳上での強度がガウス分布であるときのエネルギーの計算に相当します。

```
> f:=exp(-x^2-y^2/5);
```


$$f := e^{-x^2 - \frac{1}{5}y^2}$$

(9)

```
> p1:=plot3d(f,x=-3..3,y=-3..3,axes=boxed):
```

```
> p2:=plots[cylinderplot]([2,theta,z],z=0..1,theta=0..2*Pi,style=patchngrid,color=pink):
```

```
> plots[display](p1,p2);
```



円柱座標系に変換します。

```
> cf:=simplify(subs({x=r*cos(theta),y=r*sin(theta)},f));
```

$$cf := e^{-\frac{1}{5}r^2(4\cos^2(\theta) + 1)}$$

(10)

VectorCalculus パッケージの Jacobian コマンドを用いて、ヤコビアン行列を計算します。

```
> VectorCalculus[Jacobian]([r*cos(theta), r*sin(theta),h],[r,theta,h]);
```

(11)

$$\begin{bmatrix} \cos(\theta) & -r \sin(\theta) & 0 \\ \sin(\theta) & r \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (11)$$

行列を計算します。

```
> J:=simplify(LinearAlgebra[Determinant]((11)) );
      J:=r
```

(12)

円筒の体積(エネルギー)

```
> int(int(cf * J, r=0..2), theta=0..2 * Pi);
```

$$\int_0^{2\pi} \left(-\frac{5}{2} \frac{-1 + e^{-\frac{16}{5} \cos(\theta)^2 - \frac{4}{5}}}{4 \cos(\theta)^2 + 1} \right) d\theta$$

(13)

```
> evalf((13)) ;
```

5.300553595

(14)